/*
 * The MIT License
 *
 * Copyright (c) 2010, Marc Schwerdtfeger (marc@schwerdtfeger.de)
 *
 * Permission is hereby granted, free of charge, to any person obtaining a copy
 * of this software and associated documentation files (the "Software"), to deal
 * in the Software without restriction, including without limitation the rights
 * to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
 * copies of the Software, and to permit persons to whom the Software is
 * furnished to do so, subject to the following conditions:
 *
 * The above copyright notice and this permission notice shall be included in
 * all copies or substantial portions of the Software.
 *
 * THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
 * IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
 * FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
 * AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
 * LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
 * OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
 * THE SOFTWARE.
 */

package org.loosix.core.event
{
import flash.events.Event;
import flash.events.EventDispatcher;
import flash.utils.Dictionary;

    import org.loosix.core.lxc_internal;

    use namespace lxc_internal;

    public class TopicDispatcher
    {
        private static var _instance:TopicDispatcher; // Singleton Pattern
        private var _topicMap:Dictionary; // eine Liste mit allen Topics. F�r jedes Topic wird ein EventDispatcher erzeugt.

        /**
         * Konstruktor
         * @param useLoosix privater Konstruktor
         */
        public function TopicDispatcher(useLoosix:LoosixEnforcer)
        {
            _topicMap = new Dictionary();
        } // endConstructor

        /**
         * Registriert ein Ereignis-Listener-Objekt bei einem EventDispatcher-Objekt, sodass der Listener �ber ein
         * Ereignis benachrichtigt wird
         *
         * @param topic der Name des Topics
         * @param type der Typ des Ereignisses
         * @param listener Die Listener-Funktion, die das Ereignis verarbeitet
         * @param useCapture Legt fest, ob der Listener in der Aufnahmephase oder in der Ziel- und Bubbling-Phase arbeitet.
         * @param priority Die Priorit�tsstufe des Ereignis-Listeners. 
         * @param useWeakReference Legt fest, ob der Verweis auf den Listener stark oder schwach ist. Ein starker
         *                         Verweis (der Standard) verhindert, dass der Listener von einem Garbage Collector
         *                         entfernt wird. Ein schwacher Verweis hingegen nicht. 
         */
        public function addEventListener(topic:String, type:String, listener:Function, useCapture:Boolean = false, priority:int = 0, useWeakReference:Boolean = false):void
        {
            var dispatcher:EventDispatcher;

            if (!hasTopic(topic))
            {
                _topicMap[topic] = new EventDispatcher();
            }   // endIf

            dispatcher = _topicMap[topic];
            dispatcher.addEventListener(type, listener, useCapture, priority, useWeakReference);
        } // endFunc

        /**
         * Entfernt einen Listener aus dem EventDispatcher-Objekt. Wenn kein entsprechender Listener mit dem
         * EventDispatcher-Objekt registriert ist, hat ein Aufruf dieser Methode keine Auswirkungen.
         *
         * @param topic der Name des Topics
         * @param type der Typ des Ereignisses
         * @param listener Die Listener-Funktion, die das Ereignis verarbeitet
         * @param useCapture Legt fest, ob der Listener in der Aufnahmephase oder in der Ziel- und Bubbling-Phase arbeitet.
         */
        public function removeEventListener(topic:String, type:String, listener:Function, useCapture:Boolean = false):void
        {
            var dispatcher:EventDispatcher;

            if (hasTopic(topic))
            {
                dispatcher = _topicMap[topic];
                dispatcher.removeEventListener(type, listener, useCapture);
            }   // endIf

        } // endFunc

        /**
         *  Sendet ein Ereignis in den Ereignisablauf. Das Ereignisziel ist das EventDispatcher-Objekt,
         *  f�r das die Methode dispatchEvent() aufgerufen wird.
         *
         * @param topic der Name des Topics
         * @param event Das Ereignisobjekt, das in den Ereignisablauf integriert wird.
         */
        public function dispatchEvent(topic:String, event:Event):void
        {
            if(hasTopic(topic))
            {
                var dispatcher:EventDispatcher = _topicMap[topic];
                dispatcher.dispatchEvent(event);
            } // endIf

        } // endFunc

        /**
         * Pr�ft ob f�r das �bergebene Topic ein Eventdispatcher existiert.
         * @param topic der Name des Topics
         * @return true|false
         */
        private function hasTopic(topic:String):Boolean
        {
            return (_topicMap[topic] != null);
        } // endFunc

        lxc_internal static function getInstance():TopicDispatcher
        {
            if (_instance == null) _instance = new TopicDispatcher(new LoosixEnforcer());
            return _instance;
        } // endFunc

    } // endClass

} // endPackage
class LoosixEnforcer{}